
[bookmark: _Hlk206164841]生まれた年月日の①十干十二支（干支）②九星③十二星座を表示します。
また、毎日健康に過ごすために気を付けたいことややっておきたいことの一言を選んで表示します。

①十干十二支（じっかんじゅうにし）とは
十干は甲、乙、丙、丁、戊、己、庚、辛、壬、癸の10種類で、日を順に10日のまとまりで数えるための呼び名（符号）でした。10日ごとに、「一旬（いちじゅん）」と呼び、3つの旬（上旬、中旬、下旬）で1ヶ月になるため、広く使われていました。これに陰陽五行説をあてはめ、陽を兄（え）陰を弟（と）にして、五行の木火土金水と組み合わせて「きのえ」などと読んでいます。
十二支は子、丑、寅、卯、辰、巳、午、未、申、酉、戌、亥の12種類です。十二支にはそれぞれ動物が当てられていますが、もともと動物とは関係がなく、１２か月の順を表すものです。時刻を表すことにも使われています。
②九星（きゅうせい）とは
九星は、約４０００年前の中国の夏の創始者が思いついたのが始まりといわれ、陰陽五行説を組み合わされて現在の形になった東洋占術の一つです。日本で一般に広がったのは神宮暦に掲載されてからです。方位学、個人の運気、家相や土地の吉凶を見る際にも使われています。
③十二星座とは
地球から見た太陽の天球上での一年間の見かけの通り道である黄道帯に配置した十二の星座に対して太陽や月、水星、金星、火星、木星、土星、天王星、海王星、冥王星の10個の惑星がどのような位置関係にあるかをもとに占う西洋占星術です。個人を占う場合は、生年月日をもとにして占います。
占いで使う十二星座は実際の星座の位置でなく十二等分したものに付けた名前。

入力した生年月日に対応する十干十二支、九星、十二星座の表示と健康占いを作ります。

使う関数
１，TODAY関数（日付/時刻グループ）
今日の日付を返します。
書式　TODAY()
シートの一部を変更したとき、再計算指定をしたときに今日の日付を返します。
２．VLOOKUP関数（検索/行列グループ）
「Vertical（バーチカル）」(垂直)を意味するVと「LOOKUP（ルックアップ）」(探す)の略で、表の縦方向にデータを検索し、一致した値と同じ行にあるデータを取り出すか、検索する値以下の最大値を一致した値として同じ行にあるデータを取り出す関数です。元となる表のデータから値を取り出すため、数字や文字の間違いや表記のブレを防ぐこともできます。
書式　VLOOKUP(検索値、表（範囲）、取り出す列番号、検索方法）
表を作る時、検索する値は第１列に設定します。取り出す値は第２列以降に設定します。検索方法には、「TRUE」か「FALSE」をいれ、値が一致したとき検索できたとする場合は「FALSE」を使います。省略した場合は「TRUE」となります。
２．MOD関数（数学/三角グループ）
数値を割る数で割ったときの余り（除数）を返します。
書式　MOD(数値, 除数)
３．YEAR関数（日付/時刻グループ）
年月日の年を取り出します。
書式　YEAR(シリアル値)
シリアル値は年月日を1900年1月1日を1日目とした通算日を言います。
４．MONTH関数（日付/時刻グループ）
[bookmark: _Hlk206167602]年月日の月を取り出します。
書式　MONTH(シリアル値)
５，DAY関数（日付/時刻グループ）
年月日の日を取り出します。
書式　DAY(シリアル値)
６．RANDBETWEEN関数（数学/三角グループ）
シートの一部を変更したときまたは再計算指定をしたときに指定した最小値から最大値の範囲内の整数をランダムに返します。
書式　RANDBETWEEN (最小値、最大値)
占いを作成
１．前準備
準備された『健康占い（数式なし）』ブックを開きます。
２．『表』シートの各表に名前を付ける
表に名前を付けることで関数の引数をわかりやすくできます。
『表』シートを選びます。
①十干十二支の表に名前を付ける
ⅰ『表』シートのセル範囲［A3:D62］を選びます。
ⅱ『数式』タブの『定義された名前』グループの［名前の定義］の文字部分をクリックします。
ⅲ表示された『新しい名前』ボックスの『名前』の枠に「十干十二支」と入力します。
ⅳ［OK］ボタンをクリックします。
②九星の表に名前を付ける
[bookmark: _Hlk206169931]ⅰ『表』シートのセル範囲［E3:G11］を選びます。
ⅱ『数式』タブの『定義された名前』グループの［名前の定義］の文字部分をクリックします。
ⅲ表示された『新しい名前』ボックスの『名前』の枠に「九星」と入力します。
ⅳ［OK］ボタンをクリックします。
③十二星座の表に名前を付ける
ⅰ『表』シートのセル範囲［H3:J15］を選びます。
ⅱ『数式』タブの『定義された名前』グループの［名前の定義］の文字部分をクリックします。
ⅲ表示された『新しい名前』ボックスの『名前』の枠に「十二星座」と入力します。
ⅳ［OK］ボタンをクリックします。
④健康運の表に名前を付ける
ⅰ『表』シートのセル範囲［K3:L102］を選びます。
ⅱ『数式』タブの『定義された名前』グループの［名前の定義］の文字部分をクリックします。
ⅲ表示された『新しい名前』ボックスの『名前』の枠に「健康運」と入力します。
ⅳ［OK］ボタンをクリックします。
３．セルの表示の書式を入れる
①セル［B1］に日付書式を入れます。
ⅰセル［B1］を選び右クリックします。
ⅱ表示されたメニューの［セルの書式設定］をクリックします。
ⅲ『セルの書式設定』ボックスの『表示形式』タブの『分類』にある［日付］をクリックします。
ⅳ右に表示された『種類』にある［○月○日］をクリックします。
ⅴ［OK］ボタンをクリックします。
②セル［B2］に日付書式を入れます。
ⅰセル［B2］を選び右クリックします。
ⅱ表示されたメニューの［セルの書式設定］をクリックします。
ⅲ『セルの書式設定』ボックスの『表示形式』タブの『分類』にある［日付］をクリックします。
ⅳ右に表示された『種類』にある［（年号）○年○月○日］をクリックします。
ⅴ［OK］ボタンをクリックします。
４．『入力』シートに数式を入れる
数式を入れる前にセル［B2］に和暦を入力しておくと数式を入れることで計算の結果が表示され見やすくなります
①『今日は』の横のセル［B1］に今日の日付を表示するために次の式を入力します。
=TODAY()
②『生まれ年十干十二支』の横のセル［B3］に十干十二支を表示するために次の式を入力します。
=VLOOKUP(MOD((YEAR(B2)-1924),60),十干十二支,2,FALSE)
③『生まれ年十干十二支』の横のセル［C3］に十干十二支の日本読みを表示するために次の式を入力します。
=VLOOKUP(MOD((YEAR(B2)-1924),60),十干十二支,3,FALSE)
④『生まれ年十干十二支』の横のセル［D3］に十干十二支の音読みを表示するために次の式を入力します。
=VLOOKUP(MOD((YEAR(B2)-1924),60),十干十二支,4,FALSE)
⑤『生まれ年九星』の横のセル［B4］に九星を表示するために次の式を入力します。
=VLOOKUP(MOD(YEAR(B2)-1928,9),九星,2,FALSE)
⑥『生まれ年九星』の横のセル［C4］に九星の日本語読みを表示するために次の式を入力します。
=VLOOKUP(MOD(YEAR(B2)-1928,9),九星,3,FALSE)
[bookmark: _Hlk206182421]⑦『生まれ月日十二星座』の横のセル［B5］に十二星座を表示するために次の式を入力します。
=VLOOKUP(MONTH(B2)*100+DAY(B2),十二星座,2,TRUE)
⑧『生まれ月日十二星座』の横のセル［C5］に十二星座のラテン名を表示するために次の式を入力します。
=VLOOKUP(MONTH(B2)*100+DAY(B2),十二星座,3,TRUE)
⑨『健康占い』の横のセル［B6］に健康運を表示するために次の式を入力します。
=VLOOKUP(RANDBETWEEN(1,100),健康運,2,FALSE)
